Rules for integrands of the form $(a + b \operatorname{Log}[c x^n])^p$

1: $\int (a + b \operatorname{Log}[c x^n])^p \, dx$ when $p > 0$

- **Reference: G&R 2.711.1, CRC 485, CRC 490**
- **Derivation: Integration by parts**
- **Rule: If $p > 0$, then**

$$\int (a + b \operatorname{Log}[c x^n])^p \, dx \longrightarrow x (a + b \operatorname{Log}[c x^n])^p - b n p \int (a + b \operatorname{Log}[c x^n])^{p-1} \, dx$$

- **Program code:**

```
Int[Log[c_.*x_^n_.],x_Symbol] :=
  x*Log[c*x^n] - n*x /;
FreeQ[{c,n},x]
```

```
Int[(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  x*(a+b*Log[c*x^n])^p - b*n*p*Int[(a+b*Log[c*x^n])^(p-1),x] /;
FreeQ[{a,b,c,n},x] && GtQ[p,0] && IntegerQ[2*p]
```

2: $\int (a + b \operatorname{Log}[c x^n])^p \, dx$ when $p < -1$

- **Derivation: Inverted integration by parts**
- **Rule: If $p < -1$, then**

$$\int (a + b \operatorname{Log}[c x^n])^p \, dx \longrightarrow \frac{x (a + b \operatorname{Log}[c x^n])^{p+1}}{b n (p+1)} - \frac{1}{b n (p+1)} \int (a + b \operatorname{Log}[c x^n])^{p+1} \, dx$$

- **Program code:**

```
Int[(a_.+b_.*Log[c_.*x_^n_.])^p_,x_Symbol] :=
  x*(a+b*Log[c*x^n])^(p+1)/(b*n*(p+1)) - 1/(b*n*(p+1))*Int[(a+b*Log[c*x^n])^(p+1),x] /;
FreeQ[{a,b,c,n},x] && LtQ[p,-1] && IntegerQ[2*p]
```

3. $\int (a + b \,\mathrm{Log}[c\, x^n])^p \, dx$ when $\frac{1}{n} \in \mathbb{Z}$

1: $\int \frac{1}{\mathrm{Log}[c\, x]} \, dx$

Reference: CRC 492

Derivation: Integration by substitution and algebraic simplification

- **Basis:** $F[\mathrm{Log}[c\, x]] = \frac{1}{c} \mathrm{Subst}[e^x F[x], x, \mathrm{Log}[c\, x]] \, \partial_x \mathrm{Log}[c\, x]$
- **Basis:** $\int \frac{e^x}{x} \, dx = \mathrm{ExpIntegralEi}[x]$
- **Basis:** $\mathrm{ExpIntegralEi}[\mathrm{Log}[z]] = \mathrm{LogIntegral}[z]$

Note: This rule is optional, but returns antiderivative expressed in terms of `LogIntegral` instead of `ExpIntegralEi`.

Rule:

$$\int \frac{1}{\mathrm{Log}[c\, x]} \, dx \longrightarrow \frac{1}{c} \mathrm{Subst}\left[\int \frac{e^x}{x} \, dx, x, \mathrm{Log}[c\, x]\right] \longrightarrow \frac{1}{c} \mathrm{ExpIntegralEi}[\mathrm{Log}[c\, x]] \longrightarrow \frac{1}{c} \mathrm{LogIntegral}[c\, x]$$

Program code:

```
Int[1/Log[c_.*x_],x_Symbol] :=
  LogIntegral[c*x]/c /;
FreeQ[c,x]
```

2: $\int (a+b\,\mathrm{Log}[c\,x^n])^p\,dx$ when $\frac{1}{n}\in\mathbb{Z}$

Derivation: Integration by substitution

- **Basis: If** $\frac{1}{n}\in\mathbb{Z}$, **then** $F[\mathrm{Log}[c\,x^n]] == \frac{1}{n\,c^{1/n}}\,\mathrm{Subst}\left[e^{x/n}\,F[x],\,x,\,\mathrm{Log}[c\,x^n]\right]\,\partial_x\mathrm{Log}[c\,x^n]$
- **Rule: If** $\frac{1}{n}\in\mathbb{Z}$, **then**

$$\int (a+b\,\mathrm{Log}[c\,x^n])^p\,dx \longrightarrow \frac{1}{n\,c^{1/n}}\,\mathrm{Subst}\left[\int e^{x/n}\,(a+b\,x)^p\,dx,\,x,\,\mathrm{Log}[c\,x^n]\right]$$

$$\int (a+b\,\mathrm{Log}[c\,x^n])^p\,dx \longrightarrow \frac{1}{b\,n\,c^{1/n}\,e^{\frac{a}{b\,n}}}\,\mathrm{Subst}\left[\int x^p\,e^{\frac{x}{b\,n}}\,dx,\,x,\,a+b\,\mathrm{Log}[c\,x^n]\right]$$

- **Program code:**

```
Int[(a_.+b_.*Log[c_.*x_^n_.])^p_,x_Symbol] :=
  1/(n*c^(1/n))*Subst[Int[E^(x/n)*(a+b*x)^p,x],x,Log[c*x^n]] /;
FreeQ[{a,b,c,p},x] && IntegerQ[1/n]
```

4: $\int (a+b\,\mathrm{Log}[c\,x^n])^p\,dx$

- **Derivation: Piecewise constant extraction and integration by substitution**
- **Basis:** $\partial_x \frac{x}{(c\,x^n)^{1/n}} == 0$
- **Basis:** $\frac{(c\,x^n)^k\,F[\mathrm{Log}[c\,x^n]]}{x} == \frac{1}{n}\,\mathrm{Subst}\left[e^{k\,x}\,F[x],\,x,\,\mathrm{Log}[c\,x^n]\right]\,\partial_x\mathrm{Log}[c\,x^n]$
- **Rule:**

$$\int (a+b\,\mathrm{Log}[c\,x^n])^p\,dx \longrightarrow \frac{x}{(c\,x^n)^{1/n}}\int \frac{(c\,x^n)^{1/n}\,(a+b\,\mathrm{Log}[c\,x^n])^p}{x}\,dx \longrightarrow \frac{x}{n\,(c\,x^n)^{1/n}}\,\mathrm{Subst}\left[\int e^{x/n}\,(a+b\,x)^p\,dx,\,x,\,\mathrm{Log}[c\,x^n]\right]$$

$$\int (a+b\,\mathrm{Log}[c\,x^n])^p\,dx \longrightarrow \frac{x}{(c\,x^n)^{1/n}}\int \frac{(c\,x^n)^{1/n}\,(a+b\,\mathrm{Log}[c\,x^n])^p}{x}\,dx \longrightarrow \frac{x}{b\,n\,(c\,x^n)^{1/n}\,e^{\frac{a}{b\,n}}}\,\mathrm{Subst}\left[\int x^p\,e^{\frac{x}{b\,n}}\,dx,\,x,\,a+b\,\mathrm{Log}[c\,x^n]\right]$$

- **Program code:**

```
Int[(a_.+b_.*Log[c_.*x_^n_.])^p_,x_Symbol] :=
  x/(n*(c*x^n)^(1/n))*Subst[Int[E^(x/n)*(a+b*x)^p,x],x,Log[c*x^n]] /;
FreeQ[{a,b,c,n,p},x]
```